

A Minimalist Putback-Based Bidirectional Programming Language, Formally Verified

Hsiang-Shang Ko², Tao Zan¹, and Zhenjiang Hu^{1,2}

¹ SOKENDAI (The Graduate University for Advanced Studies), Japan
`{taozan,hu}@nii.ac.jp`

² National Institute of Informatics, Japan
`hsiang-shang@nii.ac.jp`

1 Introduction

Theory construction in terms of AGDA programming

2 Decidable partiality

The bidirectional transformations discussed in this paper are “decidably partial”. By “decidable partiality” we mean that the transformations are actually total functions which wrap their results in the standard **Maybe** datatype:

```
data Maybe (A : Set) : Set where  
  just    : A → Maybe A  
  nothing : Maybe A
```

The result of a piece of successful computation is wrapped in the **just** constructor, while failed computation is represented by **nothing**. The behaviour of decidably partial functions is different from the usual partial functions as formalised in terms of complete partial orders (see, e.g., Gibbons [5]): when defining the semantics of **BiGUL**, we need to define programs that produce some result or fail respectively when a sub-program fails or produces some result, but this is not possible in the setting of complete partial orders (since such functions violate monotonicity, a necessary condition for computability).

We can choose to write our decidably partial bidirectional transformations directly as **Maybe**-programs, but when proving their well-behavedness, we will need to analyse or establish how a piece of decidably partial computation succeeds or fails, and the task would be much easier if the computation steps leading to success or failure are reified as data. (An example of such proofs is given in the definition of lens composition $_ \leftrightarrow _$ in Section 3.) The reification starts from the following datatype **Par**, which is a deep embedding of the combinators that we will use for writing decidably partial programs:

```
data Par : Set → Set1 where  
  return      : { A : Set } → A → Par A
```

```

runPar : { A : Set } → Par A → Maybe A
runPar (return x) = just x
runPar (mx ≫= f) with runPar mx
runPar (mx ≫= f) | just x  = runPar (f x)
runPar (mx ≫= f) | nothing = nothing
runPar fail = nothing
runPar (catch mx f my) with runPar mx
runPar (catch mx f my) | just x  = runPar (f x)
runPar (catch mx f my) | nothing = runPar my
runPar (assert true then mx) = runPar mx
runPar (assert false then mx) = nothing

```

Fig. 1. Definition of the interpreter of Par to Maybe

```

_≫=_      : { A B : Set } → Par A → (A → Par B) → Par B
fail      : { A : Set } → Par A
catch     : { A B : Set } → Par A → (A → Par B) → Par B → Par B
assert_then_ : { A B : Set } → Bool → Par A → Par A

```

Informally: A value of type `Par A` represents a piece of decidable partial computation that either successfully produces a value of type `A` or fails. The first two constructors `return` and `_≫=_` represent the usual monadic operators [7,11], and `fail` indicates failed computation. We will also need error handling, which is represented by the `catch` constructor. The behaviour of our `catch` is slightly different from the usual one: execution of `catch mx f my` binds the result of `mx` to `f` if `mx` computes successfully, or goes to execution of `my` if `mx` fails to compute. Also we will frequently need to ensure that a `Bool` value is `true` before continuing with the rest of a program, so the constructor `assert_then_` is included for convenience — execution of `assert b then p` becomes execution of `p` when `b` is `true`, or fails otherwise.

Formally, the meaning of the constructors of `Par` is given by the interpreter $runPar : \{A : \text{Set}\} \rightarrow \text{Par } A \rightarrow \text{Maybe } A$ shown in Figure 1. We say that $mx : \text{Par } A$ computes to $x : A$ exactly when $runPar \ mx \equiv \text{just } x$, and that mx fails to compute exactly when $runPar \ mx \equiv \text{nothing}$. The first equality is what we aim to reify as data next.

Our reification of decidable partial computation as the datatype `Par` and AGDA’s support of full dependent types allow us to compute for any $mx : \text{Par } A$ and $x : A$ a type $mx \mapsto x$ whose inhabitants explain how mx computes to x step by step. Below is a sample of the definition of $_ \mapsto _$:

```

_↦_ : { A : Set } → Par A → A → Set1
return x ↦ x' = x ≡ x'
mx ≫= f ↦ y = (x : _) × (mx ↦ x) × (f x ↦ y)
...

```

That is, `return x` computes to x' when x is exactly x' , and $mx \gg= f$ computes to y when there exists $x : A$ such that mx computes to x and $f \ x$ computes to y

(and so on). Here our use of the term “computes to” is justified, since we can prove that

$$mx \mapsto x \quad \text{and} \quad \text{runPar } mx \equiv \text{just } x$$

are equivalent. That is, a term of type $mx \mapsto x$ can be constructed if and only if $\text{runPar } mx \equiv \text{just } x$, meaning that our reification of $\text{runPar } mx \equiv \text{just } x$ as $mx \mapsto x$ is accurate. We are thus entitled to use $\mapsto$ for stating various properties involving successful decidable partial computation, in particular well-behavedness of bidirectional transformations in the next section.

3 Bidirectional transformations formalised

We follow the classic (well-behaved) lens approach [3] to bidirectional transformations, but adopt the variant used by Pacheco et al [8]:

Definition 1 (lenses). *A (well-behaved) lens between a source type $S : \text{Set}$ and a view type $V : \text{Set}$ is defined in AGDA as the following record type:*

```
record Lens (S V : Set) : Set1 where
  field
    put : S → V → Par S
    get : S → Par V
    PutGet : {s s' : S} {v : V} → (put s v ↦ s') → (get s' ↦ v)
    GetPut : {s : S} {v : V} → (get s ↦ v) → (put s v ↦ s)
```

Informally: Execution of $\text{put } s \ v$ should produce an updated version of the source s by properly embedding all information of the view v into s . The *PutGet* law ensures that put does its job properly by requiring that, after put updates a source with a view, get can completely recover the view from the updated source. The *GetPut* law, on the other hand, ensures that put does nothing more than its designated job by requiring that put performs no update on a source if the view it receives is directly extracted from the source by get . Note that our definition of lenses is stronger than the one given by Foster et al [3]: our *PutGet* law says that successful computation of put guarantees success of the subsequent computation of get , whereas in Foster et al’s definition there is no such guarantee; the *GetPut* law is similar.

A related notion is “partial” isomorphisms as defined by

```
record Iso (A B : Set) : Set1 where
  field
    to   : A → Par B
    from : B → Par A
    to-from-inverse : {x : A} {y : B} → (to x ↦ y) → (from y ↦ x)
    from-to-inverse : {x : A} {y : B} → (from y ↦ x) → (to x ↦ y)
```

`Lens` is a generalisation of `Iso` because we can easily convert `Iso A B` to `Lens A B`, whose `put` directly produces a new source from its view argument, ignoring its source argument:

```
iso-lens : { A B : Set } → Iso A B → Lens A B
iso-lens iso = record
  { put = λ _ b → Iso.from iso b
  ; get = Iso.to iso
  ; PutGet = Iso.from-to-inverse iso
  ; GetPut = Iso.to-from-inverse iso }
```

This conversion gives us a first way to manufacture lenses. For example, the identity isomorphism between the same type — whose `to` and `from` are both `return` — gives rise to the “replacing” lens, and the empty isomorphism between any two types — whose `to` and `from` are both `λ _ → fail` — gives rise to the “failure” lens. Both lenses will appear in Section 4.1.

To give the reader a taste of how well-behavedness proofs are constructed in our AGDA setting, let us try to define an operator `_↔_` for lens composition:

```
_↔_ : { A B C : Set } → Lens A B → Lens B C → Lens A C
l ↔ r = record
  { put = λ a c → Lens.get l a >>= λ b → Lens.put r b c >>= Lens.put l a
  ; get = λ a → Lens.get l a >>= Lens.get r
  ; PutGet = ?
  ; GetPut = ? }
```

The `put` and `get` functions for a composite lens can be pretty straightforwardly constructed from the `put` and `get` functions of the component lenses. As part of the definition of `_↔_`, we also need to prove that `PutGet` and `GetPut` hold for the pair of `put` and `get` we provide. For `PutGet` we need to show that, for all $a, a' : A$, and $c : C$, from a proof of $put\ a\ c \mapsto a'$ we can construct a proof of $get\ a' \mapsto c$. That is, we need to write a function converting an inhabitant of the type $put\ a\ c \mapsto a'$ to one of the type $get\ a' \mapsto c$. AGDA automatically expands the two types following the definition of `_↔_`:

$$\begin{aligned} & put\ a\ c \mapsto a' \\ = & (b : B) \times (Lens.get\ l\ a \mapsto b) \times \\ & (b' : B) \times (Lens.put\ r\ b\ c \mapsto b') \times (Lens.put\ l\ a\ b' \mapsto a') \end{aligned}$$

and

$$\begin{aligned} & get\ a' \mapsto c \\ = & (b : B) \times (Lens.get\ l\ a' \mapsto b) \times (Lens.get\ r\ b \mapsto c) \end{aligned}$$

Thus all we need to do is convert a five-tuple to a three-tuple of the above types. This is easy: Applying `Lens.PutGet l` to the fifth component of type `Lens.put l a b' ↦ a'` we obtain an inhabitant of type `Lens.get l a' b'`, and applying `Lens.PutGet r` to the fourth component of type `Lens.put r b c ↦ b'`

we obtain an inhabitant of type $\text{Lens.get } r \ b' \ c$. Hence the proof term for *PutGet* is simply

$$\lambda (b, x, b', y, z) \rightarrow (b', \text{Lens.PutGet } l \ z, \text{Lens.PutGet } r \ y)$$

With a similar reasoning, we can also construct a proof term for *GetPut*:

$$\lambda (b, x, y) \rightarrow (b, x, b, \text{Lens.GetPut } r \ y, \text{Lens.GetPut } l \ x)$$

completing the definition of $_ \leftrightarrow _$. For the proofs in the rest of this paper we will merely provide a sketch, but these proofs all have their formal counterparts in AGDA as illustrated above, completely guaranteeing their validity.

4 BiGUL and its lens semantics

Various bidirectional programming languages have been built based on the idea of lenses. Such languages would contain a set of primitive lenses, like those we build by *iso-lens*, and a set of combinators which combine smaller lenses into larger ones, like what $_ \leftrightarrow _$ does. Usually these languages are designed to suggest that what they describe are *get* programs, but it has recently been argued [2,6,8,9] that the opposite *putback-based* approach should be adopted instead. The argument is based on the following theorem:

Theorem 1 (putback is the essence). *Define, for every $mx, my : \text{Par } A$, a type $mx \approx my$ to mean that $mx \mapsto z$ if and only if $my \mapsto z$ for any $z : A$. Then for any $S, V : \text{Set}$ and two lenses $l, r : \text{Lens } S \ V$ such that*

$$\text{Lens.put } l \ s \ v \approx \text{Lens.put } r \ s \ v \quad \text{for all } s : S \text{ and } v : V,$$

we have

$$\text{Lens.get } l \ s \approx \text{Lens.get } r \ s \quad \text{for all } s : S.$$

The theorem tells us that, when designing a bidirectional combinator, if we can justify the design of its *put* component, then its *get* component needs no further justification, since there is no other *get* that can be paired with the same *put*. This is in sharp contrast with the *get*-based approach, with which the designer needs to choose for each *get* one or a few possible *put* strategies to be paired with the *get*. This complicates the design as there would be multiple combinators with the same *get* semantics, distinguished only by differences in their *put* semantics — this is rather counterintuitive as the programmer is supposed to think in terms of *get* instead of *put* when programming in *get*-based languages. Also it may well happen that, after the programmer describes a *get*, the corresponding *put* synthesised by the language is not what the programmer expects. All these problems simply do not exist for the putback-based approach.

Following the putback-based approach, our language BiGUL describes *put* functions, i.e., updates to a source using a view. Figure 2 shows (a simplified version of) the top-level definition of BiGUL. In general, a BiGUL update

```

data BiGUL : U → U → Set1 where
  replace : {S : U} → BiGUL S S
  fail    : {S V : U} → BiGUL S V
  skip    : {S : U} → BiGUL S one
  caseS   : {S V : U} → List (CaseSBranch' S V) → BiGUL S V
  caseV   : {S V : U} → List (CaseVBranch' S V) → BiGUL S V
  align   : {S V : U} → (source-condition : [ S ] → Par Bool)
                        (match : [ S ] → [ V ] → Par Bool)
                        (b : BiGUL S V)
                        (create : [ V ] → Par [ S ])
                        (conceal : [ S ] → Par (Maybe [ S ])) → BiGUL (list S) (list V)
  update  : {S : U} → (pat : Pattern S) (bs : BiGULs pat) → BiGUL S (Views' pat bs)
  rearr   : {S V V' : U} → (pat : Pattern V) (pat' : Pattern V')
                        (paths : Paths pat pat') → BiGUL S V' → BiGUL S V

```

Fig. 2. Simplified definition of BiGUL (auxiliary definitions omitted)

proceeds by decomposing the source into several parts to be updated independently (**update**; Section 4.5), and accordingly rearranging the view to match with these parts (**rearr**; Section 4.6), until the source and the view are reduced to the extent that basic operations can be applied (**replace**, **fail**, and **skip**; Section 4.1). When both the source and the view are lists, there is a powerful aligning operation for synchronising the two lists (**align**; Section 4.4). And we can do case analyses on either the source (**caseS**; Section 4.2) or the view (**caseV**; Section 4.3) for describing more flexible update strategies.

The types of the BiGUL constructors are indexed by the source and view types expected by the operations represented by the constructors (details in Section 4.5), and in the end we can write a well-typed interpreter

$$\text{interp} : \{S V : U\} \rightarrow \text{BiGUL } S V \rightarrow \text{Lens } [S] [V]$$

Due to space restrictions, we will not present the full definitions of *interp* nor the auxiliary types used in Figure 2, which are actually straightforward deep embedding of BiGUL’s lens semantics. Instead, we will only provide a sketch of the lens semantics for each of the bidirectional operations. For each operation we will start from a natural semantics (for *put*) and then add necessary checks so a corresponding *get* can exist, arriving at the lens semantics for the operation.

4.1 Replacing, failure, and skipping

The three most basic operations are **replace**, **fail**, and **skip**. The **replace** operation is applicable when the source and the view are of the same type, and its semantics is discarding the original source and returning the view as the updated source. As mentioned in Section 3, we can in fact derive its lens semantics from the identity isomorphism:

$$\begin{aligned} \text{replace-lens} &: \{S : \text{Set}\} \rightarrow \text{Lens } S \ S \\ \text{replace-lens} &= \text{iso-lens } (\text{record } \{ \text{to} = \text{return} ; \text{from} = \text{return} ; \dots \}) \end{aligned}$$

The *get* semantics for **replace** is therefore just the identity function. Also mentioned in Section 3 is that we can derive the lens semantics for **fail**, which fails to compute for any input, from the empty isomorphism in the same way:

$$\begin{aligned} \text{fail-lens} &: \{S \ V : \text{Set}\} \rightarrow \text{Lens } S \ V \\ \text{fail-lens} &= \text{iso-lens } (\text{record } \{ \text{to} = \text{fail} ; \text{from} = \text{fail} ; \dots \}) \end{aligned}$$

Its *get* semantics also fails for any input. On the other hand, the semantics of the **skip** operation is to discard the view and return the original source, and its lens semantics is not an instance of *iso-lens*:

$$\begin{aligned} \text{skip-lens} &: \{S : \text{Set}\} \rightarrow \text{Lens } S \ \top \\ \text{skip-lens} &= \text{record } \{ \text{put} = \lambda s _ \rightarrow \text{return } s ; \text{get} = \lambda _ \rightarrow \text{return } \text{tt} ; \dots \} \end{aligned}$$

Note that the view type is specified as $\top$, whose inhabitant can only be tt — had the view type been a more complex type, there would have been no way for *get* to decide what to return. In general, *put* must embed all view information into the updated source so *get* can recover the entire view from the updated source, establishing *PutGet*. $\top$ is a type with no information, and hence $\text{Lens.put skip-lens}$ can safely ignore its view argument of type $\top$.

4.2 Case analysis on source

The next construct **caseS** is for doing case analysis on the source. The basic idea of **caseS**'s semantics is to choose from a list of branches the first one that matches the source, and then execute the BiGUL statement of that branch. It is, however, difficult to find a *get* to pair with this *put* semantics: the obvious *get* which selects the first branch that matches its source argument does not work, because it may well happen that the updated source produced by *put* matches a different branch from the original source, and hence in general we would have to guarantee *PutGet* for *put* and *get* coming respectively from any two branches, which is unmanageable. A more manageable solution is to insert a dynamic check at the end of *put* to ensure that the updated source must match the same branch as the original source, which is the solution adopted by Foster et al's concrete conditional lenses [3, Section 6.1]. This solution, however, is too restrictive to be useful in practice.

Our solution is to enhance **caseS** by adding a special kind of branches called *adaptive* branches, in addition to *normal* branches. The lens semantics of the two kinds of branches are defined by the following datatype:

$$\begin{aligned} \text{data CaseSBranchType } (S \ V : \text{Set}) &: \text{Set}_1 \text{ where} \\ \text{normal} &: \text{Lens } S \ V \rightarrow \text{CaseSBranchType } S \ V \\ \text{adaptive} &: (S \rightarrow \text{Par } S) \rightarrow \text{CaseSBranchType } S \ V \end{aligned}$$

The lens semantics of a **normal** branch is just a lens, while for an **adaptive** branch it is a source transformation. With each branch we also need to associate a decidable predicate on the source type specifying when a source matches the branch, so the complete definition of the type of branches is

$$\begin{aligned} \text{CaseSBranch} &: \text{Set} \rightarrow \text{Set} \rightarrow \text{Set}_1 \\ \text{CaseSBranch } S \ V &= (S \rightarrow \text{Par Bool}) \times \text{CaseSBranchType } S \ V \end{aligned}$$

The type of **caseS**'s lens semantics *caseS-lens* is then

$$\text{caseS-lens} : (S \ V : \text{Set}) \rightarrow \text{List } (\text{CaseSBranch } S \ V) \rightarrow \text{Lens } S \ V$$

We might think of a source case analysis as still consisting mainly of normal branches, but we can now specify additional cases such that when a source matches none of the normal branches, the source can still be transformed, or adapted, to one that matches a normal branch. More explicitly, the case analysis may be run twice: if a normal branch is chosen the first time, then the case analysis succeeds and we execute the branch; otherwise, we adapt the source and rerun the case analysis, and must match a normal branch this time. After a normal branch is executed, we must ensure that the updated source satisfies the predicate of the branch and also none of the predicates of the previous branches, so a subsequent execution of *get* on the updated source would choose the same branch as *put*, establishing *PutGet*.

4.3 Case analysis on view

For **caseV**, basically we still match the view with a list of branches, and execute the first branch matched. We must be more careful when it comes to the view, though: As we mentioned in Section 4.1, view information must be carefully managed and completely embedded into the source. When the view matches a branch, the amount of view information that remains to be embedded into the source can actually be reduced. For example, if there is a branch that matches an integer view which equals zero, then for this branch we may even perform no update on the source as long as we can guarantee that the subsequent *get* will choose this branch, in which case *get* can simply return zero — the information that the view is zero is embedded implicitly in the control flow rather than explicitly in the updated source. Thus, in the definition of **CaseVBranch** below, we use a partial isomorphism on the view type which converts it to a less informative type, instead of just a decidable predicate as in **CaseSBranch**:

$$\begin{aligned} \text{CaseVBranch} &: \text{Set} \rightarrow \text{Set} \rightarrow \text{Set}_1 \\ \text{CaseVBranch } S \ V &= (V' : \text{Set}) \times \text{Lens } S \ V' \times \text{Iso } V' \ V \end{aligned}$$

For the comparison-with-zero example above, we would set V' to be $\top$, indicating that there is no more information to be embedded into the view after a successful matching, and set the *to* and *from* components of the isomorphism to $\lambda _ \rightarrow \text{return } 0$ and $\lambda \ n \rightarrow \text{assert } n == 0 \text{ then return tt}$. A branch is considered

matched with a view when the *from* conversion of the associated isomorphism succeeds on the view.

The lens semantics we assign to `caseV` has type

$$\text{caseV-lens} : (S \ V : \text{Set}) \rightarrow \text{DecEq } V \rightarrow \text{List CaseVBranch} \rightarrow \text{Lens } S \ V$$

whose `DecEq V` argument will be explained shortly. The main behaviour of its *put* is to try to match the view with each branch and execute the first successful branch. For *get*, however, there is no obvious way to decide which branch to go, as we are not given a view to start with like *put*. It seems that we can only try to execute each of the branches until we reach one that computes a view successfully, which is the approach taken by Pacheco et al [8]. This makes guaranteeing *PutGet* difficult, as we cannot easily guarantee that executing *put* followed by *get* would go through the same branch. Our compromised solution is to make *put* do an expensive check after a branch is matched and executed, requiring that the *get* for each of the previous branches must either fail or produce the same view as the input one (i.e., verifying *PutGet* directly). That is, the ranges of the branches have to be “more or less” disjoint in order for *put* to compute successfully. The `DecEq V` argument of *caseV-lens*, which expands to $(v \ v' : V) \rightarrow (v \equiv v') \uplus \neg (v \equiv v')$ and says that *V* is equipped with a decidable equality, is needed for this check.

4.4 List alignment

4.5 Source update

A fundamental operation is performing updates to parts of the source. We follow the approach taken by FLUX [1], which requires that updates must be independent — that is, the same part of a source cannot be updated more than once — to avoid the problem of sequencing updates. It turns out that a pattern matching notation suits this task perfectly: pattern matching is arguably the most intuitive way to deconstruct a source, and we can specify apparently independent updates by writing them at the variable positions in a pattern. The idea itself is straightforward; what follows, despite its slight complexity (especially if the reader is not familiar with dependently typed programming), is merely our strongly typed implementation of the idea in AGDA.

Since AGDA is fully dependently typed, we can naturally define patterns in a type-directed way such that nonsensical patterns are syntactically ruled out. This task starts from the construction of a *universe* `U`, i.e., a datatype whose inhabitants are interpreted as types. `U` has appeared in the definition of `BiGUL` in Figure 2, where it is used as the type of the indices representing the source and view types of the operations. The actual AGDA implementation of `BiGUL` uses a universe capable of expressing mutually inductive datatypes, but, to make the presentation easier to follow, the universe `U` that we define below — along

with its interpreter $\llbracket _ \rrbracket$ — is only a simplified version:

data $U : \text{Set}_1$ where	$\llbracket _ \rrbracket : U \rightarrow \text{Set}$
$k : (A : \text{Set}) \rightarrow \text{DecEq } A \rightarrow U$	$\llbracket k \ A \ _ \rrbracket = A$
$\text{one} : U$	$\llbracket \text{one} \rrbracket = \top$
$_ \oplus _ : U \rightarrow U \rightarrow U$	$\llbracket F \oplus G \rrbracket = \llbracket F \rrbracket \uplus \llbracket G \rrbracket$
$_ \otimes _ : U \rightarrow U \rightarrow U$	$\llbracket F \otimes G \rrbracket = \llbracket F \rrbracket \times \llbracket G \rrbracket$
$\text{list} : U \rightarrow U \rightarrow U$	$\llbracket \text{list } F \rrbracket = \text{List } \llbracket F \rrbracket$

The types that we can encode within U are those expressible in terms of any types with a decidable equality, $\top$, $_ \oplus _$, $_ \otimes _$, and List ; applying $\llbracket _ \rrbracket$ to an inhabitant of the universe decodes it to the type it represents. All encoded types are equipped with a decidable equality, as we can define a function of type

$$U\text{-dec} : (D : U) \rightarrow \text{DecEq } \llbracket D \rrbracket$$

by induction on D .

We can now define a family of types $\text{Pattern} : U \rightarrow \text{Set}$ such that, for any $D : U$, the type $\text{Pattern } D$ contains exactly those patterns sensible for $\llbracket D \rrbracket$:

data $\text{Pattern} : U \rightarrow \text{Set}$ **where**

- $\text{var} : \{D : U\} \rightarrow \text{Pattern } D$
- $\text{const} : \{D : U\} \rightarrow \llbracket D \rrbracket \rightarrow \text{Pattern } D$
- $\text{left} : \{D \ E : U\} \rightarrow \text{Pattern } D \rightarrow \text{Pattern } (D \oplus E)$
- $\text{right} : \{D \ E : U\} \rightarrow \text{Pattern } E \rightarrow \text{Pattern } (D \oplus E)$
- $\text{prod} : \{D \ E : U\} \rightarrow \text{Pattern } D \rightarrow \text{Pattern } E \rightarrow \text{Pattern } (D \otimes E)$
- $\text{cons} : \{D : U\} \rightarrow \text{Pattern } D \rightarrow \text{Pattern } (\text{list } D) \rightarrow \text{Pattern } (\text{list } D)$

On all types we can use **var** patterns, which match anything, and **const** patterns, which match inhabitants equal to the specified constant (this is possible because of $U\text{-dec}$); on sum types we can use **left** or **right** patterns matching inj_1 and inj_2 constructors; on product types we can use **prod** patterns to decompose a pair to its two components; on list types we can use **cons** patterns to decompose a non-empty list to its head and tail.

Next we will give patterns a different interpretation as “containers”, and store values — whose types depend on the types of the variable patterns — at their variable positions. The types of such containers can be defined by induction on Pattern :

$\text{VarPositions} : \{lv : \text{Level}\} \{D : U\} \rightarrow \text{Pattern } D \rightarrow (U \rightarrow \text{Set } lv) \rightarrow \text{Set } lv$

- $\text{VarPositions } \{D\} \ \text{var} = f \ D$
- $\text{VarPositions } (\text{const } _) = \top$
- $\text{VarPositions } (\text{left } pat) = \text{VarPositions } pat$
- $\text{VarPositions } (\text{right } pat) = \text{VarPositions } pat$
- $\text{VarPositions } (\text{prod } lpat \ rpat) = \text{VarPositions } lpat \times \text{VarPositions } rpat$
- $\text{VarPositions } (\text{cons } hpat \ tpat) = \text{VarPositions } hpat \times \text{VarPositions } tpat$

(In fact, all entities below that depend on `Pattern` are defined by a similar induction.) For example, the result of a successful pattern matching can be filled into such a container, by putting the resulting components at their respective variable positions. We thus define the types of pattern matching results as an instance of `VarPositions`:

$$\begin{aligned} \text{PatResult} &: \{D : \mathbf{U}\} \rightarrow \text{Pattern } D \rightarrow \mathbf{U} \\ \text{PatResult } pat &= \text{VarPositions } pat \llbracket _ \rrbracket \end{aligned}$$

Pattern matching is then simply a partial isomorphism:

$$pat\text{-}iso : \{D : \mathbf{U}\} \rightarrow \text{Pattern } D \rightarrow \text{Iso } \llbracket D \rrbracket (\text{PatResult } pat)$$

The *to* component of the isomorphism performs pattern matching (which can fail because matching with *left* and *right* patterns may not succeed), and the *from* component reverses pattern matching by evaluating a pattern as if it is an expression (which always succeeds), using the input of type `PatResult pat` as an environment for values at variable positions.

Similarly, by specialising `VarPositions` we can place lenses (and their view types) at the variable positions of a pattern:

$$\begin{aligned} \text{Lenses} &: \{D : \mathbf{U}\} \rightarrow \text{Pattern } D \rightarrow \text{Set}_1 \\ \text{Lenses } pat &= \text{VarPositions } pat (\lambda D \rightarrow (E : \mathbf{U}) \times \text{Lens } \llbracket D \rrbracket \llbracket E \rrbracket) \end{aligned}$$

Given a pattern *pat* and a collection of lenses *ls* : `Lenses pat` at its variable positions, we can define a lens between `PatResult pat` and a type `Views pat ls` that combines all the view types used by the lenses:

$$\begin{aligned} update\text{-}lens &: \{D : \mathbf{U}\} (pat : \text{Pattern } D) (ls : \text{Lenses } pat) \rightarrow \\ &\quad \text{Lens } (\text{PatResult } pat) (\text{Views } pat \text{ } ls) \end{aligned}$$

where the definition of `Views` is as straightforward as

$$\begin{aligned} \text{Views} &: \{D : \mathbf{U}\} (pat : \text{Pattern } D) \rightarrow \text{Lenses } pat \rightarrow \text{Set} \\ \text{Views } \{D\} \text{ var } (E, _) &= \llbracket E \rrbracket \\ \text{Views } (\text{const } _) \text{ } ls &= \top \\ \text{Views } (\text{left } pat) \text{ } ls &= \text{Views } pat \text{ } ls \\ \text{Views } (\text{right } pat) \text{ } ls &= \text{Views } pat \text{ } ls \\ \text{Views } (\text{prod } lpat \text{ } rpat) (ls, ls') &= \text{Views } lpat \text{ } ls \times \text{Views } rpat \text{ } ls' \\ \text{Views } (\text{cons } hpat \text{ } tpat) (ls, ls') &= \text{Views } hpat \text{ } ls \times \text{Views } tpat \text{ } ls' \end{aligned}$$

The lens semantics of the `update` operation is then just the composition of *pat-iso* and *update-lens*. Its *put* decomposes the source, updates its components with the lenses at the corresponding positions, and putting the updated components together, while its *get* again decomposes the source and extracts views from its components with the corresponding lenses.

4.6 View rearrangement

The last operation **rearr** is for rearranging the view to a specific shape suitable for subsequent updates. For example, the **update** operation in Section 4.5 demands that the view must be in a shape that matches the source pattern, and this is usually achieved by an outer **rearr**. In essence, **rearr** performs a piece of invertible computation (which loses no information), while its *get* semantics performs the inverse of the computation. The invertible computation used is usually not complicated (hence the name “rearrangement”), so instead of letting the programmer write general invertible functions (perhaps along with their inverses), we ask instead for syntactic descriptions of simple invertible computation, which are more restrictive but allow automatic inversion.

The kind of invertible computation we wish to express is a decomposition of an old view followed by an assembling of a new view using all components of the old view. When it comes to decomposition, pattern matching should come to mind again; assembling can also be handled by pattern matching, as we have formulated pattern matching as an isomorphism in Section 4.5. We thus require two patterns $pat : \text{Pattern } D$ and $pat' : \text{Pattern } E$, the former for decomposing the old view of type $\llbracket D \rrbracket$ and the latter for assembling the new view of type $\llbracket E \rrbracket$. For the latter pattern pat' we also need to specify which component of the old view will be used as the value at each variable position. This last piece of information is represented by specialising **VarPositions** again:

$$\begin{aligned} \text{Paths} &: \{ D E : \mathcal{U} \} \rightarrow \text{Pattern } D \rightarrow \text{Pattern } E \rightarrow \text{Set}_1 \\ \text{Paths } pat \ pat' &= \text{VarPositions } pat' \ (\text{Path } pat) \end{aligned}$$

where $\text{Path } pat \ T$ is the type of all paths that lead to a component of type $\llbracket T \rrbracket$ at a variable position of a **PatResult** pat :

$$\begin{aligned} \text{Path} &: \{ D : \mathcal{U} \} \rightarrow \text{Pattern } D \rightarrow \mathcal{U} \rightarrow \text{Set}_1 \\ \text{Path } \{ D \} \text{ var} & \quad T = D \equiv T \\ \text{Path } (k \ _) & \quad T = \perp \\ \text{Path } (\text{left } pat) & \quad T = \text{Path } pat \ T \\ \text{Path } (\text{right } pat) & \quad T = \text{Path } pat \ T \\ \text{Path } (\text{prod } lpat \ rpat) & \quad T = \text{Path } lpat \ T \uplus \text{Path } rpat \ T \\ \text{Path } (\text{cons } hpat \ tpat) & \quad T = \text{Path } hpat \ T \uplus \text{Path } tpat \ T \end{aligned}$$

If the $paths : \text{Paths } pat \ pat'$ supplied uses up all of the old view, i.e., all possible paths for pat are contained in $paths$, then pat , pat' , and $paths$ together specify a piece of computation that can be automatically inverted — since all components of the old view are present somewhere in the new view, we can always reassemble the old view from the new view (unless there is inconsistency — a component of the old view might be copied into more than one positions of the new view, and the values at these positions of the new view must be the same when doing the reassembling). In short, we can construct an isomorphism:

$$\begin{aligned} &\text{view-rearrangement-iso} : \\ &\{ D : \mathcal{U} \} (pat : \text{Pattern } D) \{ E : \mathcal{U} \} (pat' : \text{Pattern } E) \rightarrow \\ &(\text{paths} : \text{Paths } pat \ pat') \rightarrow \text{Invertible } pat \ pat' \ paths \rightarrow \text{Iso } \llbracket E \rrbracket \llbracket D \rrbracket \end{aligned}$$

where Invertible $pat\ pat'\ paths$ is defined to state that $paths$ uses up all of the old view. The lens semantics we assign to $rearr$ is the lens derived from this isomorphism (by *iso-lens*) composed with the lens semantics of the inner BiGUL operation.

5 Examples

address book for align, update, and rearrangement

BiYACC for source and view cases (probably not)

source/view if, map in terms of align

6 Discussion

recursion (the lack thereof)

extensions: for example, iteration, view dependency

design/programming phase: design phase needs to consider *put* and *get* together, but programming phase is about *put* only

Haskell implementation

Conditions for *put* to succeed; deferred as we will have a dependently typed version (a completely formal approach to total lenses)

stems from work on BiFLUX; comparison with putlenses

Theory construction in terms of AGDA programming
Regarding formalisation: “[W]e are interested in extending what can be calculated precisely because we are not interested in the calculations themselves. Developing techniques that allow more of a derivation to be calculated allows attention to be focussed on the creative parts, which cannot be calculated.” [4]
[10]

References

1. James Cheney. FLUX: functional updates for XML. In *International Conference on Functional Programming*, ICFP’08, pages 3–14. ACM, 2008.
2. Sebastian Fischer, Zhenjiang Hu, and Hugo Pacheco. The essence of bidirectional programming. *SCIENCE CHINA Information Sciences*, 58(5):1–21, 2015.
3. J. Nathan Foster, Michael B. Greenwald, Jonathan T. Moore, Benjamin C. Pierce, and Alan Schmitt. Combinators for bidirectional tree transformations: A linguistic approach to the view-update problem. *ACM Transactions on Programming Languages and Systems*, 29(3):17, 2007.
4. Jeremy Gibbons. An introduction to the Bird-Meertens formalism. In *New Zealand Formal Program Development Colloquium*, pages 1–12, 1994.

5. Jeremy Gibbons. Calculating functional programs. In *Algebraic and Coalgebraic Methods in the Mathematics of Program Construction*, number 2297 in Lecture Notes in Computer Science, chapter 5, pages 149–201. Springer-Verlag, 2002.
6. Zhenjiang Hu, Hugo Pacheco, and Sebastian Fischer. Validity checking of putback transformations in bidirectional programming. In *International Symposium on Formal Methods*, volume 8442 of *Lecture Notes in Computer Science*, pages 1–15. Springer-Verlag, 2014.
7. Eugenio Moggi. Notions of computation and monads. *Information and Computation*, 93(1):55–92, 1991.
8. Hugo Pacheco, Zhenjiang Hu, and Sebastian Fischer. Monadic combinators for “putback” style bidirectional programming. In *Partial Evaluation and Program Manipulation*, PEPM’14, pages 39–50. ACM, 2014.
9. Hugo Pacheco, Tao Zan, and Zhenjiang Hu. BiFluX: A bidirectional functional update language for XML. In *Principles and Practice of Declarative Programming*, PPDP’14, pages 147–158. ACM, 2014.
10. The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Institute for Advanced Study, Princeton, 2013.
11. Philip Wadler. The essence of functional programming. In *Principles of Programming Languages*, POPL’92, pages 1–14. ACM, 1992.

Appendix: AGDA syntax

An AGDA datatype is defined by listing all of its constructors along with their types in full. A datatype can be *parametrised*, as in the case of `Maybe`, which has a parameter $A : \text{Set}$, specified between the name of the datatype and the colon on the first line. (`Set` is the type of all (small) types.) Parameters can be referred to throughout a datatype definition.

Here the datatype `Par` is *indexed* by `Set` (which appears to the right of the colon on the first line), meaning that `Par` is in fact a `Set`-indexed family of simultaneously defined types.

As for the constructors, let us look at the simplest case, `fail`: The constructor `fail` constructs a value of type `Par A` for any $A : \text{Set}$. There is thus one argument to `fail`, namely $A : \text{Set}$, and the type of `fail`'s result depends on the value of this argument (which happens to be a type). This is an example of *dependent function types*, whose result type can depend on the value of the argument. Logically, dependent function types are read as universally quantified propositions — the dependent function type $\{x : A\} \rightarrow B$ is read as “for all $x : A$ the property B holds for x ”. (Ordinary function types, on the other hand, denote implications logically.)

Back to `fail`: most of the time AGDA can infer what `fail`'s argument should be based on how `fail` is used, so we make the argument *implicit* by enclosing it in curly brackets. (Explicit arguments, on the other hand, are enclosed in round brackets.) We can then invoke `fail` without supplying the argument, leaving it for AGDA to infer. We de-emphasise implicit arguments typographically when they cause visual annoyance, as in this case.

An AGDA name can be used as an infix (or, in general, mixfix) operator when the name contains underscores, in which case the underscores indicate where the arguments should go. For example, we can apply `_>>=_` to arguments mx and f by writing either `_>>= mx f` or `mx >>= f`.

The type $\top : \text{Set}$ has exactly one constructor `tt` : $\top$, while the type $\perp : \text{Set}$ has no inhabitant — logically they are read as truth and falsity respectively. The type family `_≡_` : $\{A : \text{Set}\} \rightarrow A \rightarrow A \rightarrow \text{Set}$ is defined such that $x \equiv y$ is inhabited if and only if x and y are equal. The type former `_⊔_` : $\text{Set} \rightarrow \text{Set} \rightarrow \text{Set}$ is disjoint union (disjunction) whose constructors are `inj1` : $\{A : \text{Set}\} \rightarrow A \rightarrow A \uplus B$ and `inj2` : $\{A : \text{Set}\} \rightarrow B \rightarrow A \uplus B$, and the type former `_×_` : $\text{Set} \rightarrow \text{Set} \rightarrow \text{Set}$ is cartesian product (conjunction), whose constructor is `_,_`. The symbol ‘ $\times$ ’ is also overloaded to denote *dependent pair types* — the inhabitants of a dependent pair type $(x : A) \times B$ are pairs where the type of the second component depends on the value of the first component. Logically, dependent pair types are read as existentially quantified propositions — $(x : A) \times B$ is read as “there exists $x : A$ such that the property B holds for x ”.

An underscore in a right-hand side expression instructs AGDA to infer what should be placed there.

An AGDA record is defined by listing the names and types of all its components, and a component type can depend on previous components. To access a component of an inhabitant of a record type, we should use the corresponding

component extraction function. For example, $\text{Lens.get} : \{S \ V : \text{Set}\} \rightarrow \text{Lens } S \ V \rightarrow S \rightarrow \text{Par } V$ is for extracting the *get* component of a **Lens**.